

Chapter

08

## 尋找與參照函數

8-1 LOOKUP函數基本應用

8-2 MATCH 與 INDEX 函數應用

## 8-1 LOOKUP 函數基本應用

LOOKUP 函數家族共有 VLOOKUP、HLOOKUP 與 LOOKUP 函數，它們是非常重要的，而且能協助我們迅速尋找到所要的資料。此節先說明其基本的應用方式。

### 8-1-1 VLOOKUP 函數

VLOOKUP 函數用在某一陣列資料範圍中，由使用者設定最左欄的特定值，去尋找同列中所指定欄位的特定內容。

語法：VLOOKUP(lookup\_value,table\_array,col\_index\_num,range\_lookup)

由使用者設定最左欄的特定值，去尋找同列中所指定欄位的特定內容。

[ 引數 ]：

- lookup\_value** 在陣列 ( 資料範圍 ) 的最左欄中，所要搜尋的特定值。可以是數值、參照位址或文字字串。
- table\_array** 所要搜尋的資料範圍。一般而言是指儲存格範圍的參照位址或範圍名稱。table\_array 第一欄裡的值，可以是文字、數字或邏輯值，其中字母的大小寫將被視為是相同的。
- col\_index\_num** 是個數值，為欲尋找欄位的序號，也就是要傳回的特定值位於 table\_array 中的欄。如果 col\_index\_num 小 1，則 VLOOKUP 傳回錯誤值 #VALUE!；如果 col\_index\_num 超過 table\_array 總欄數，則 VLOOKUP 傳回錯誤值 #REF!。

`range_lookup` 是個邏輯值，用來指定 `VLOOKUP` 是要尋找完全符合或部分符合的值。當此引數值為 `TRUE` 或被省略，則會傳回部分符合的值；也就是說，如果找不到完全符合的值時，會傳回僅次於 `lookup_value` 的值。當此引數值為 `FALSE` 時，`VLOOKUP` 函數只會尋找完全符合的值，如果找不到，則傳回錯誤資訊 `#N/A`。

### 提示

如果 `range_lookup` 邏輯值為 `TRUE`，則 `table_array` 第一欄的值必須以遞增次序排列：...、-2、-1、0、1、2、...、A-Z、`FALSE`、`TRUE`；否則 `VLOOKUP` 將會出現錯誤的資料。我們可以點選 資料 > 排序 指令，讓其依遞增 順序排列。如果 `range_lookup` 邏輯值為 `FALSE`，則 `table_array` 不須事先排序。

### 範例

#### VLOOKUP 函數基本應用

**Step01** 開啟 `ch08.xls` 範例檔案，選定 8-1-1 工作表。


**Step02** 此範例資料範圍為 `D2:F7`，是所得稅率表。

	A	B	C	D	E	F	G	H
1				年所得下限	年所得上限	所得稅率		
2	請輸入年所得	\$123,838		\$0	\$2,650	15.00%		
3	所得稅率	36.00%		\$2,651	\$27,300	28.00%		
4	應繳稅款	\$44,581.68		\$27,301	\$58,500	31.00%		
5				\$58,501	\$131,800	36.00%		
6				\$131,801	\$284,700	39.60%		
7				\$284,701		45.25%		
8								

**Step03** 在 B2 儲存格輸入一年的總所得，例如：123838。

**Step04** 在 B3 儲存格輸入以下公式，或使用 函數精靈 處理，即可找到對應的所得稅率。

=VLOOKUP(B2,D2:F7,3)



### 提示

公式中，B2 為特定值，D2:F7 為資料範圍，3 是指第 3 欄資料，邏輯值省略代表是 TRUE。

## 8-1-2 HLOOKUP 函數

HLOOKUP 函數用在某一陣列資料範圍中，由使用者設定最上方一列的特定值，去尋找同欄位中指定某列的特定內容。

語法：HLOOKUP(lookup\_value,table\_array,row\_index\_num,range\_lookup)

由使用者設定最上方一列的特定值，去尋找同欄位中指定某列的特定內容。

[ 引數 ]：

lookup\_valu 在陣列(資料範圍)中第一列，所要搜尋的特定值。可以是數值、參照位址或文字字串。

- table\_array** 所要搜尋的資料範圍。一般而言是指儲存格範圍的參照位址或範圍名稱。table\_array 第一列中的值，可以是文字、數字或邏輯值，其中字母的大小寫將被視為是相同的。
- row\_index\_num** 是個數值，為欲尋找列的序號，也就是要傳回的特定值位於 table\_array 列中的第幾列。如果 row\_index\_num 小於 1，則 HLOOKUP 函數傳回錯誤值 #VALUE!；如果 row\_index\_num 超過 table\_array 的總列數，則 HLOOKUP 函數傳回錯誤值 #REF!。
- range\_lookup** 是個邏輯值，用來指定 HLOOKUP 是要尋找完全符合或部分符合的值。當此引數值為 TRUE 或被省略了，則會傳回部分符合的值；也就是說，如果找不到完全符合的值時，會傳回僅次於 lookup\_value 的值。當此引數值為 FALSE 時，HLOOKUP 函數只會尋找完全符合的值，如果找不到，則傳回錯誤資訊 #N/A。

### 提示


如果 range\_lookup 邏輯值為 TRUE，則 table\_array 第一列的值必須以遞增次序排列：...、-2、-1、0、1、2、...、A-Z、FALSE、TRUE；否則 HLOOKUP 將會出現錯誤的資料。我們可以點選 資料 > 排序 指令，讓其依遞增 順序排列。如果 range\_lookup 邏輯值為 FALSE，則 table\_array 不須事先排序。

### 範例

#### 使用 HLOOKUP 函數

**Step01** 開啟 ch08.xls 範例檔案，選定 8-1-2 工作表。


**Step02** 此範例資料範圍 D2:J4，是所得稅率表。



**Step03** 在 B2 儲存格輸入一年的總所得，例如：23838。

**Step04** 在 B3 儲存格輸入以下公式，或使用 函數精靈 協助輸入，即可找到對應的所得稅率。

=HLOOKUP(B2,D2:J4,3)



提示

公式中，B2 為特定值，D2:J4 為資料範圍，3 是指第 3 列資料，邏輯值省略代表是 TRUE。

## 8-1-3 LOOKUP 函數

LOOKUP 函數有 向量語法型式 與 陣列語法型式 二種，其目的是一樣的，都是利用指定的搜尋值，在另一個儲存格範圍中找到對應的資料。

### || 向量型式

語法：LOOKUP(lookup\_value,lookup\_vector,result\_vector)

利用指定的搜尋值，在另一個儲存格範圍中找到對應的資料。

[ 引數 ]：

- lookup\_valu      函數在向量所要尋找的特定值。lookup\_value 可以是數字、文字、邏輯值或參照到數值的名稱或參照位址。
- lookup\_vector    是個單列或單欄的範圍。在 lookup\_vector 中的值可能是文字、數字或邏輯值。
- result\_vecto      是個單列或單欄的範圍。它的大小應與 lookup\_vector 相同。

#### 提示

在 lookup\_vector 中的數值必須以遞增次序排列：…，-2，-1，0，1，2，…，A-Z，FALSE，TRUE；否則，LOOKUP 不會傳回正確的值。字母的大小寫被視為是相同的。

#### 範例

#### 向量型式 LOOKUP 函數

**Step01** 開啟 ch08.xls 範例檔案，選定 8-1-3 工作表。



**Step02** 此範例中的 D2:D7 是尋找範圍，F2:F7 是結果範圍。

**Step03** 在 B2 輸入一年的年所得，例如：33888。



**Step04** 在 B3 儲存格輸入公式，或使用 函數精靈 協助輸入，找到答案為 31%。

=LOOKUP(B2,D2:D7,F2:F7)





## 陣列型式

語法：LOOKUP(lookup\_value,array)

利用指定的搜尋值，在另一個儲存格範圍中找到對應的資料。

[ 引數 ]：

**lookup\_value** 函數在陣列中所要尋找的特定值。lookup\_value 可以是個數字、文字、邏輯值，或是參照到數值的名稱或參照位址。如果 LOOKUP 函數無法找 lookup\_value，它會採用陣列中等於或小於 lookup\_vector 的最大值。如果 lookup\_value 小於第一列或第一欄(依照陣列維數決定)中最小值時，則 LOOKUP 函數傳回錯誤值#N/A。Array 是一個儲存格範圍，其中含有所要與 lookup\_value 比較的文字、數字或邏輯值。

### 提示

在 lookup\_vector 中的數值必須以遞增次序排列：...，-2，-1，0，1，2，...，A-Z，FALSE，TRUE；否則，LOOKUP 不會傳回正確的值。字母的大小寫被視為是相同的。

依循前例，在 B6 儲存格輸入下列公式，找到答案為 31%。

=LOOKUP(B2,D3:F4)

	A	B	C	D	E	F	G
1				年所得下限	年所得上限	所得稅率	
2	請輸入年所得	\$33,888		\$0	\$2,650	15.00%	
3	所得稅率	31%		\$2,651	\$27,300	28.00%	
4	應繳稅款	\$10,505.28		\$27,301	\$58,500	31.00%	
5				\$58,501	\$131,800	36.00%	
6	陣列型式	0.31		\$131,801	\$284,700	39.60%	
7				\$284,701		45.25%	
8							
9							

## 8-2 MATCH 與 INDEX 函數應用

### 8-2-1 尋找左側欄位資料

一般在使用 VLOOKUP 函數值，只能尋找特定值右側的資料，如果要尋找左側的資料，就會用到 LOOKUP 函數，不過，您可能會發現所找到的資料經常不精確。為了解決這個問題，建議您可以使用 INDEX 與 MATCH 函數搭配，即能尋找到正確的左側資料。

語法：MATCH(lookup\_value,lookup\_array,match\_type)

根據指定的比對方式，傳回一陣列中與搜尋值相符合之相對位置。

[ 引數 ]：

**lookup\_value** 要在資料範圍中尋找的比對值。例如：在電話簿中尋找朋友的電話號碼時，姓名就是所要尋找的比對值，而電話號碼則是所要的資料。lookup\_value 可以是數字、文字、邏輯值，或是一個參照到數字、文字、邏輯值的參照位址。

**lookup\_array** 一連續的儲存格範圍，其中含有被比對值的資料。lookup\_array 必須是陣列或是陣列參照。

**match\_type** 數值，其值有三種選擇：-1、0 或 1。如果 match\_type 引數被省略，則假設其值為 1。

1：函數會找到等於或僅次於 lookup\_value 的值。lookup\_array 必須以遞增次序排列：...-2,-1,0,1,2,..., A-Z, FALSE, TRUE。

0：函數會找第一個完全等於 lookup\_value 的比較值。lookup\_array 可以依任意次序排列。

-1：函數會找到等於或大於 lookup\_value 的最小值。lookup\_array 必須以遞減次序排序：TRUE,FALSE,Z-A,...2, 1, 0, -1, -2,...，以此類推。

語法 (陣列) : INDEX(array,row\_num,column\_num)

擷取依據指定欄列號碼所決定的表格或陣列中一個元素的值。

[ 引數 ] :

array	陣列式的儲存格範圍。如果陣列只包含單一的列或欄時，則所對應的 row_num 或 column_num 是可省略的。如果陣列含有多列多欄的元素，卻只單獨使用 row_num 或 column_num，則函數將以陣列形式傳回陣列中的某一整列或整欄元素。
row_num	指定所要的元素是位於陣列裡的第幾列。如果省略了 row_num 這個引數，則一定要輸入 column_num。
column_num	指定所要傳回的元素是位於陣列裡的第幾欄。如省略 column_num 這個引數，則一定要輸入 row_num。

範例

在表格中尋找左側欄位資料

**Step01** 開啟 ch08.xls 範例，選定 8-2-1 工作表。

**Step02** 此範例 A1:C15 是資料範圍。

**Step03** 在 E1 輸入特定值，例如：郭硬漢。

**Step04** 在 E6 輸入下列公式，以便查詢得到其打擊數，得到答案為 70 個打擊數目。

```
=INDEX(A1:A15,MATCH(E1,C1:C15))
```

**Step05** 在 F7 輸入下列公式，以便查詢其打擊率，得到其打擊率為 0.227。

```
=INDEX(B1:B15,MATCH(E1,C1:C15))
```



	A	B	C	D	E	F
1	打數	打擊率	球員	球員姓名: 郭硬漢		
2	88	0.414	澎恰恰	打擊率	=LOOKUP(\$E\$1,球員,打擊率)	<<==== LOOKUP
3	115	0.393	章太燿			
4	47	0.349	陳制遠	打數	=LOOKUP(\$E\$1,球員,打數)	<<==== LOOKUP
5	58	0.288	蔡風毅			
6	38	0.263	陳今風	打數	=INDEX(打擊率,MATCH(\$E\$1,球員,0))	<<==== INDEX and
7	77	0.26	劉甘霖	打擊率	=INDEX(打擊率,MATCH(\$E\$1,球員,0))	<<==== INDEX and
8	20	0.25	林明賢			
9	92	0.244	許怪手			
10	73	0.229	徐聲明			
11	70	0.227	郭硬漢			
12	45	0.227	王不壯			
13	65	0.18	羅閻氫			
14	80	0.177	高惟缸			
15	96	0.169	睇重章			

顯示公式

	A	B	C	D	E	F	G	H
1	打數	打擊率	球員	球員姓名: 郭硬漢				
2	88	0.414	澎恰恰	打擊率	0.177	<<==== LOOKUP		
3	115	0.393	章太燿					
4	47	0.349	陳制遠	打數	80	<<==== LOOKUP		
5	58	0.288	蔡風毅					
6	38	0.263	陳今風	打數	70	<<==== INDEX and MATCH		
7	77	0.260	劉甘霖	打擊率	0.227	<<==== INDEX and MATCH		
8	20	0.250	林明賢					
9	92	0.244	許怪手					
10	73	0.229	徐聲明					
11	70	0.227	郭硬漢					
12	45	0.227	王不壯					
13	65	0.180	羅閻氫					
14	80	0.177	高惟缸					
15	96	0.169	睇重章					

顯示結果

## 提示

您可參考此範例 E2 與 E4 儲存格，它使用 LOOKUP 函數查詢，結果其所查得的資料是不正確的，所以此時使用 Index 與 Match 函數，較為恰當。

## 8-2-2 雙向搜尋

前面所提到的搜尋方式，都是由一個特定值出發，在資料範圍中尋找所要的對應值。此小節的範例，將說明由縱、橫二個特定值，來尋找資料範圍的對應值。

範例

### 雙向搜尋

**Step01** 開啟 ch08.xls，選定 8-2-2 工作表。

**Step02** 此範例中 D1:H14 為資料範圍。

**Step03** 如果要查詢七月的雜誌銷售量，請在 B1 儲存格輸入 七月，B2 儲存格輸入 雜誌。

**Step04** 在 B9 儲存格輸入下列公式，求得答案為 1300。

```
=INDEX(E1:G13,MATCH(B1,D2:D13,0),MATCH(B2,E1:G1,0))
```

	A	B	C	D	E	F	G	H
1	月份：	七月			圖書	雜誌	數位學習	小計
2	產品：	雜誌		一月	2,850	1,500	3,215	7,565
3				二月	3,650	1,900	2,561	8,111
4	月份對應儲存格：	7		三月	5,680	1,820	4,826	12,326
5	產品對應儲存格：	2		四月	4,220	1,650	1,108	6,978
6	銷售量：	1,300		五月	3,530	1,900	2,023	7,453
7				六月	1,728	1,400	2,965	6,093
8				七月	5,185	1,300	3,100	9,585
9	單一公式====>	1,300		八月	3,256	1,500	2,610	7,366
10				九月	1,800	1,100	2,050	4,950
11				十月	2,548	1,600	3,650	7,798
12				十一月	4,086	1,200	5,320	10,606
13				十二月	3,684	1,500	2,635	7,819
14				月小計	42,217	18,370	36,063	96,650

## 8-2-3 尋找線性內插值

在執行一些計算工作時，對於某些線性的圖表數據，可能在某些重要節點，並沒有對應的資料，此時可以利用 INDEX 與 LOOKUP 函數配合，來求得任二點之間的特定內插數值。

範例

### 尋找線性內插值

**Step01** 開啟 ch08.xls 範例檔案，選定 8-2-3 工作表。

**Step02** 範例中 D2:D14 為資料範圍，想求得 X=3 的時候，Y 值的對應數值。

**Step03** 在 B1 儲存格輸入 3。

**Step04** 在 B3 儲存格輸入下列公式，得到 X=3，在資料範圍所對應的列數。  
=LOOKUP(B1,D2:D14,D2:D14)

**Step05** 在 B4 儲存格輸入下列公式，確認 X=3 是否剛好在已知數值節點上。  
=B1=B3

**Step06** 在 B6 儲存格輸入下列公式，求得內插值起始的列數。  
=MATCH(B3,D2:D14,0)

**Step07** 在 B7 儲存格輸入下列公式，求得內插值的終點的列數。  
=IF(B4,B6,B6+1)

**Step08** 在 B9 儲存格輸入下列公式，求得內插值的起始 X 數值。  
=INDEX(D2:D14,B6)

**Step09** 在 B10 儲存格輸入下列公式，求得內插值的終止 X 數值。  
=INDEX(D2:D14,B7)

**Step10** 在 B12 儲存格輸入下列公式，求得內插值起始 Y 數值。  
=LOOKUP(B9,D2:D14,E2:E14)

**Step11** 在 B13 儲存格輸入下列公式，求得內插值終點 Y 數值。


=LOOKUP(B10,D2:D14,E2:E14)

**Step12** 在 B15 儲存格輸入下列公式，求得內插調整比例。

=IF(B4,0,(B1-B3)/B10-B9))

**Step13** 在 B16 儲存格輸入下列公式，求得 X=3 的內插值 Y=21。

=B12+((B13-B12)\*B15)



結果